Computation of the perimeter of a spherical uniform random disc-polygon

Notation: κ for the curvature, κg for the geodesic curvature, κg' and κg'' for its first and second derivatives, respectively,
and $\Delta\kappa$ for κg-cot(r)

Implicit differentiation to obtain t in terms of x

The starting function is $C(x) - K(x)$ as described in the paper.
The value of $G_n[0, 0]$ is the n-th derivative at $t = 0$ of the function representing (in terms of x) the implicit function $C(x) - K(x) = 0$.

```
In[1]:= G[x_, t_] := Cos[r] * Sin[r + t] - Cos[r + t] * Sqrt[(Sin[r])^2 - x^2] -
          (κg * x^2 / 2 + κg' * x^3 / 6 + (κg'' + 3 * κg + 3 κg^3) x^4 / 24)

In[2]:= G1[x_, t_] := -D[G[x, t], x] / D[G[x, t], t]
        G1[x, t] /. {x → 0, t → 0}

Out[3]= 0

In[10]:= G2[x_, t_] := D[G1[x, t], x] + D[G1[x, t], t] * G1[x, t]
         Assuming[r ≤ Pi / 2 && r > 0 && t < r, Simplify[G2[x, t] /. {x → 0, t → 0}]]

Out[11]=
         κg - Cot[r]

In[12]:= G3[x_, t_] := D[G2[x, t], x] + D[G2[x, t], t] * G1[x, t]
         Assuming[r ≤ Pi / 2 && r > 0, Simplify[G3[x, t] /. {x → 0, t → 0}]]

Out[13]=
         κg′

In[14]:= G4[x_, t_] := D[G3[x, t], x] + D[G3[x, t], t] * G1[x, t]
         a = Assuming[r ≤ Pi / 2 && r > 0, Simplify[G4[x, t] /. {x → 0, t → 0, k^2 → kg^2 + 1}]]

Out[15]=
```

$$\frac{3}{2}\left(6\,\kappa g + 2\,\kappa g^3 - 3\,\mathrm{Cot}[r]\,\mathrm{Csc}[r]^2 + \mathrm{Cos}[3\,r]\,\mathrm{Csc}[r]^3\right) + \kappa g''$$

We summarize the coefficients of the series expansion here (already including the factorials), and introduce the new notation $\Delta\kappa$g = κg - cot(r):

```
In[16]:= d2 := Δκg / 2
         d3 := κg' / 6
         d4 := (κg'' + 9 * Δκg / (Sin[r])^2 + 9 * Δκg^2 * Cot[r] + 3 * Δκg^3) / 24
```

We simplified the fourth coefficient manually, check that it is correct:

```
In[19]:= Simplify[(a - d4 * 24) /. {Δκg → κg - Cot[r]}]

Out[19]=
         0
```

Coefficients of the inverse i.e. obtaining x in terms of t

In[20]:=
```
C1 = 1 / (d2) ^ (1 / 2)
C2 = Simplify[-d3 / (2 * d2^2)]
C3 = Simplify[-d4 / (2 * d2^ (5 / 2)) + 5 * d3^2 / (8 * d2^ (7 / 2))]
```

Out[20]=
$$\frac{\sqrt{2}}{\sqrt{\Delta\kappa g}}$$

Out[21]=
$$-\frac{\kappa g'}{3\,\Delta\kappa g^2}$$

Out[22]=
$$\frac{5\,(\kappa g')^2 - 3\,\Delta\kappa g\,\left(3\,\Delta\kappa g\,\left(\Delta\kappa g^2 + 3\,\Delta\kappa g\,\mathrm{Cot}[r] + 3\,\mathrm{Csc}[r]^2\right) + \kappa g''\right)}{18\,\sqrt{2}\,\Delta\kappa g^{7/2}}$$

The x coordinates of the intersection points are then

In[23]:=
```
xm[t_] := -C1 * t^ (1 / 2) + C2 * t - C3 * t^ (3 / 2)
xp[t_] := C1 * t^ (1 / 2) + C2 * t + C3 * t^ (3 / 2)
```

Central angle

In[25]:=
```
θ := Collect[Simplify[
    2 * C1 * t^ (1 / 2) / Sin[r] + 2 / Sin[r] * (C3 + C1^3 / (6 * (Sin[r]) ^2)) * t^ (3 / 2)], t]
ϑ1 = Coefficient[θ, t^ (1 / 2)]
ϑ2 = Coefficient[θ, t^ (3 / 2)]
```

Out[26]=
$$\frac{2\,\sqrt{2}\,\mathrm{Csc}[r]}{\sqrt{\Delta\kappa g}}$$

Out[27]=
$$-\frac{\sqrt{\Delta\kappa g}\,\mathrm{Csc}[r]}{\sqrt{2}} - \frac{3\,\mathrm{Cot}[r]\times\mathrm{Csc}[r]}{\sqrt{2}\,\sqrt{\Delta\kappa g}} - \frac{5\,\mathrm{Csc}[r]^3}{3\,\sqrt{2}\,\Delta\kappa g^{3/2}} + \frac{5\,\mathrm{Csc}[r]\,(\kappa g')^2}{9\,\sqrt{2}\,\Delta\kappa g^{7/2}} - \frac{\mathrm{Csc}[r]\,\kappa g''}{3\,\sqrt{2}\,\Delta\kappa g^{5/2}}$$

Surface Area

In[28]:=
```
Unprotect[C]
```

In[29]:=
```
C[x_] := t + x^2 / 2 * (Cot[r] - t) + x^4 / 8 * (Cot[r] - t) / (Sin[r]) ^2
K[x_] := x^2 / 2 * κg + x^3 / 6 * κg' + x^4 / 24 * (κg'' + 3 * κg^3 + 3 * κg)
```

After integrating with respect to y, we collect the coefficients, specifically x^2 and x^4

In[31]:=
```
e2[t_] := Coefficient[
   Collect[Simplify[Series[Series[Collect[Expand[Integrate[1 + x^2 / 2 + 3 * x^4 / 8 +
           y^2 / 2 + 3 * y^4 / 8, {y, K[x], C[x]}]], x], {x, 0, 4}], {t, 0, 2}]], x], x^2]
Collect[Simplify[e2[t]], t]
```

Out[32]=
$$\frac{1}{4}\,t^2\,\mathrm{Cot}[r] + \frac{1}{4}\,(-2\,\kappa g + 2\,\mathrm{Cot}[r])$$

In[33]:=
```
e4[t_] := Coefficient[
  Collect[Simplify[Series[Series[Collect[Expand[Integrate[1 + x^2 / 2 + 3 * x^4 / 8 +
        y^2 / 2 + 3 * y^4 / 8, {y, K[x], C[x]}]], x], {x, 0, 4}], {t, 0, 2}]], x], x^4]
Collect[Simplify[-24 * e4[t]], t]
```

Out[34]=

$$3\,\kappa g\,\left(3+\kappa g^2\right)-6\,\mathrm{Cot}[r]+t^2\left(\frac{9\,\mathrm{Cot}[r]}{2}-\frac{3\,\mathrm{Cot}[r]^3}{2}\right)-3\,\mathrm{Cot}[r]\,\mathrm{Csc}[r]^2+\kappa g''$$

Which by a manual simplification yields

In[35]:=
```
κg'' + 9 * Δκg * ((Cot[r])^2 + 1) + Δκg^2 * 9 * Cot[r] +
 3 * Δκg^3 + 3 / 2 t^2 Cot[r] (2 - 2 Cot[r]^2 + Csc[r]^2)
```

Out[35]=

$$3\,\Delta\kappa g^3+9\,\Delta\kappa g^2\,\mathrm{Cot}[r]+9\,\Delta\kappa g\,\left(1+\mathrm{Cot}[r]^2\right)+\frac{3}{2}\,t^2\,\mathrm{Cot}[r]\,\left(2-2\,\mathrm{Cot}[r]^2+\mathrm{Csc}[r]^2\right)+\kappa g''$$

This is in fact correct:

In[36]:=
```
Simplify[
 Collect[Simplify[-24 * e4[t]], t] - (κg'' + 9 * Δκg * ((Cot[r])^2 + 1) + Δκg^2 * 9 * Cot[r] +
     3 * Δκg^3 + 3 / 2 t^2 Cot[r] (2 - 2 Cot[r]^2 + Csc[r]^2)) /. Δκg → κg - Cot[r]]
```

Out[36]=

0

We integrate the significant terms:

In[37]:=
```
Int[t_, x_] := t - x^2 / 2 * Δκg - x^3 / 6 * κg' -
   x^4 / 24 * (κg'' + 9 * Δκg * ((Cot[r])^2 + 1) + Δκg^2 * 9 * Cot[r] + 3 * Δκg^3)
v1 = Collect[Expand[Coefficient[Integrate[Int[t, x], {x, xm[t], xp[t]}], t, 3 / 2]], Δκg]
v2 = Collect[Expand[Coefficient[Integrate[Int[t, x], {x, xm[t], xp[t]}], t, 5 / 2]], Δκg]
```

Out[38]=

$$\frac{4\,\sqrt{2}}{3\,\sqrt{\Delta\kappa g}}$$

Out[39]=

$$-\frac{1}{5}\,\sqrt{2}\,\sqrt{\Delta\kappa g}-\frac{3\,\sqrt{2}\,\mathrm{Cot}[r]}{5\,\sqrt{\Delta\kappa g}}+\frac{-\frac{3\,\sqrt{2}}{5}-\frac{3}{5}\,\sqrt{2}\,\mathrm{Cot}[r]^2}{\Delta\kappa g^{3/2}}+\frac{\sqrt{2}\,\left(\kappa g'\right)^2}{9\,\Delta\kappa g^{7/2}}-\frac{\sqrt{2}\,\kappa g''}{15\,\Delta\kappa g^{5/2}}$$

The final formula

In[40]:=
```
D1 := Δκg * ϑ1^4
D1D2 := ϑ1^4 * (1 / (Sin[r])^2 + Δκg * Cot[r]) + Δκg * (4 * ϑ1^3 * ϑ2 - ϑ1^6 / 15)
```

The expression in the integral is

In[42]:=
```
Ihat = Collect[Simplify[
    Expand[(Sin[r])^4 * A^(2 / 3) * (D1 * v2 * (8 / 3) * (5 / 3) * Gamma[5 / 3] / v1^(5 / 3) -
        D1D2 * (5 / 3) * Gamma[5 / 3] / v1^(2 / 3)) / (18 * v1^2)]], Δκg]
```

Out[42]=

$$\frac{1}{2}\left(\frac{3}{2}\right)^{2/3}A^{2/3}\,\Delta\kappa g^{4/3}\,\mathrm{Gamma}\left[\frac{5}{3}\right]+\left(\frac{2}{3}\right)^{1/3}A^{2/3}\,\Delta\kappa g^{1/3}\,\mathrm{Cot}[r]\times\mathrm{Gamma}\left[\frac{5}{3}\right]-\frac{5\,A^{2/3}\,\mathrm{Gamma}\left[\frac{5}{3}\right]\,\left(\kappa g'\right)^2}{6\times 2^{2/3}\times 3^{1/3}\,\Delta\kappa g^{8/3}}+\frac{A^{2/3}\,\mathrm{Gamma}\left[\frac{5}{3}\right]\,\kappa g''}{2\times 2^{2/3}\times 3^{1/3}\,\Delta\kappa g^{5/3}}$$

In[43]:=
```
Ihat = Collect[Simplify[Expand[(Sin[r])^4 * A^(2/3) * (5/3) *
      Gamma[5/3] (D1 * v2 * (8/3) - D1D2 * v1) / (18 * v1^(11/3))]], Δκg]
```

Out[43]=

$$\frac{1}{2}\left(\frac{3}{2}\right)^{2/3} A^{2/3}\, \Delta\kappa g^{4/3}\, \mathrm{Gamma}\left[\frac{5}{3}\right] + \left(\frac{2}{3}\right)^{1/3} A^{2/3}\, \Delta\kappa g^{1/3}\, \mathrm{Cot}[r] \times \mathrm{Gamma}\left[\frac{5}{3}\right] - \frac{5\, A^{2/3}\, \mathrm{Gamma}\left[\frac{5}{3}\right] \left(\kappa g'\right)^2}{6 \times 2^{2/3} \times 3^{1/3}\, \Delta\kappa g^{8/3}} + \frac{A^{2/3}\, \mathrm{Gamma}\left[\frac{5}{3}\right] \kappa g''}{2 \times 2^{2/3} \times 3^{1/3}\, \Delta\kappa g^{5/3}}$$

We factor out the following constant:

In[44]:=
```
c = Gamma[5/3] * (2/3)^(1/3) * A^(2/3)
```

Out[44]=

$$\left(\frac{2}{3}\right)^{1/3} A^{2/3}\, \mathrm{Gamma}\left[\frac{5}{3}\right]$$

and obtain

In[45]:=
```
Collect[Simplify[Expand[Ihat / c]], Δκg]
```

Out[45]=

$$\frac{3\, \Delta\kappa g^{4/3}}{4} + \Delta\kappa g^{1/3}\, \mathrm{Cot}[r] - \frac{5\, \left(\kappa g'\right)^2}{12\, \Delta\kappa g^{8/3}} + \frac{\kappa g''}{4\, \Delta\kappa g^{5/3}}$$